



NETAPP TECHNICAL DOCUMENT

Symantec Enterprise Vault 7.x Cascading NetApp Snapshots

Peter Henneberry, Global Systems Engineer - NetApp
September 2008 | TR-3709

TABLE OF CONTENTS

- 1 CASCADING SNAPSHOT COPY 3**
- 2 BACKGROUND 3**
 - 2.1 BACKING UP EV DATA 4
 - SCRIPT NEEDS TO ACCOMPLISH: 5
 - 2.2 OBTAINING THE BATCH FILE TEMPLATE 6
 - 2.2.1 *Network Time Protocol (NTP)* 6
 - 2.2.2 *Configuring SSH* 6
 - 2.2.3 *Preparing each Servers* 9
 - 2.2.3.1 Enterprise Vault Server(s) 9
 - 2.2.3.2 Microsoft SQL Server(s) 10
 - 2.2.4 *Using EVNTAP.BAT for Cascading Snapshots* 10
- 3 RESTORING EV TO A PREVIOUS SNAPSHOT 10**
- 4 APPENDIX A – EVNTAP.BAT 11**
- 5 BEST PRACTICES 19**
- 6 CONCLUSION 19**

1 CASCADING SNAPSHOT COPY

This document covers the Symantec Enterprise Vault® server techniques for backing up Enterprise Vault data, as well as the ability to restore it while providing data consistency and a reduced backup/recovery window. The procedure is based on a NetApp disk-based backup and recovery solution, known as a NetApp Snapshot. The paper does not discuss the data management tools involved with the process, such as SnapDrive® for Windows®, SnapManager for SQL® Server, NetApp Snapshot®, and SnapRestore® technologies. For details on the best practice when configuring Enterprise Vault and Microsoft SQL server with NetApp storage, as well as Open Systems SnapVault® for each Enterprise Vault server.

http://media.netapp.com/documents/tr_3635.pdf


<http://media.netapp.com/documents/tr-3466.pdf>

This paper is not a replacement to any Enterprise Vault product manuals. This paper is also not applicable for designing an Enterprise Vault implementation architecture. If data components are stored on non-NetApp storage systems, explained procedures will not work.

2 BACKGROUND

Symantec Enterprise Vault deployments require a well-designed architecture to address data availability, performance, and reliability. Enterprise Vault has several moving parts in reference to its underlying data sets. Enterprise Vault application needs to coordinate with Microsoft SQL Server and various client services. It is a challenging task to back up the various components of Enterprise Vault data sets while maintaining data consistency. Each component of an Enterprise Vault data set plays an important role. It is a common practice to rebuild the Enterprise Vault Index in case of corruption of that Index data. Rebuilding the Index could take days, affecting not only the data availability, but also the performance of Enterprise Vault server. This might result in lost productivity. This paper provides a solution to address such issues by creating the backup with cascading Snapshot copies. Creating Snapshot copies happens almost instantly. Therefore, by employing this strategy, Enterprise Vault server will be placed in a read-only mode for a very brief period, then returning Enterprise Vault to a read-write mode. By using this paper, an effective disaster recovery solution may be developed by using NetApp storage management solutions..

The backup procedure to maintain data consistency without the cascading Snapshot copies involves a complex and time consuming alternative process. By using the cascading Snapshot copy approach, an Enterprise Vault administrator can configure an elegant solution for backup and recovery requirements.



2.1 BACKING UP EV DATA

Normally it is required to stop all Enterprise Vault services prior to backing up the data. Stopping Enterprise Vault services allows you to maintain data consistency among various Enterprise Vault data components such as Enterprise Vault databases, vault store databases, indexes, shopping service, and archival process. This paper provides a procedure to perform backup of data while Enterprise Vault server is online. This procedure helps to improve data availability compared to offline backup processes.

Additionally, as data is removed from Microsoft Exchange® following the EV archive, as soon as the safety copy is removed, implementing NetApp cascading Snapshots protects the archives. Normally a lost archive may only be recovered through the recovery of Microsoft Exchange mailbox data, but with the NetApp cascading Snapshot solution, a corrupted archive may be recovered directly from the Microsoft SQL server through the NetApp SnapManager for SQL application – this reduces the time to recover as well as simplifies the recovery procedure.

Components that should be backed up are Exchange, SQL system data and EV databases for all EV components (Directory, EV Vault Stores, Discovery Accelerator, Reporting, Indexes, etc), as well as EV system state and local file systems. Also any staging areas for PST migrations, shopping, or collections should be protected. SQL and EV coordination are the most critical and the backup should occur ideally after a successful Exchange backup.

Note: To ensure the best protection and recovery times, Enterprise Vault must be placed in backup mode (read-only mode) across all Enterprise Vault systems in a given site during the NetApp cascading Snapshots.

During the backup window, EV archiving will not be possible so NetApp Snapshots should be scheduled at times that do not overlap with EV Archiving runs. Traditional backup solutions modify the archive bit of the Vault Store data which triggers EV to remove the safety copy of data. With NetApp, a trigger must be used to initiate the removal of the archive bit, which is covered in the script file located in Appendix A.

The most common method used in Enterprise configurations for Enterprise Vault is to schedule scripts to run that will perform the following tasks:

SCRIPT NEEDS TO ACCOMPLISH:

- a. Stop EV services
- b. Place EV in backup mode on all EV servers within a site
- c. Start EV services
- d. Take a NetApp Snapshot from all SQL and EV servers
- e. Stop EV services
- f. Place EV in Normal mode
- g. Start EV services
- h. Execute trigger to remove safety copy after verification of successful backup
- i. Replicate the NetApp Snapshot copies on secondary storage for DR or tape retention

Put Enterprise Vault into Read-Only Mode

This is to ensure data consistency between different sets of data within the Enterprise Vault environment. It is accomplished by changing the registry key values. During this mode, users will still be able to access e-mails without having the ability to restore items from the archive.

Use SnapManager for SQL Server Backup/Restore CLI-Based (command line interface) commands to backup SQL Server and Enterprise Vault related databases

Maintaining the database in a consistent state is critical in an Enterprise Vault environment. SnapManager for SQL Server provides the tool to back up SQL Server databases, quiescing the data first to ensure it is captured in a consistent state..

Use CLI based commands to back up Enterprise Vault Archived items (Snapshot copies of network share). Replicate the Enterprise Vault archived items to another NetApp storage system using SnapMirror to back up the data.

Release Enterprise Vault from Read-Only Mode

It is important to release Enterprise Vault server into read-write mode. During read-only mode, items will continue to be queued. Releasing it enables it to restart the archival process.

Removing the safety copy depends on the settings selected based on your environment. With NetApp, a trigger must be used to initiate the removal of the archive bit, which is covered in the script file located in Appendix A.

<http://seer.entsupport.symantec.com/docs/273151.htm>

2.2 OBTAINING THE BATCH FILE TEMPLATE

Although you may download a batch file from Symantec Support Web site (<http://support.veritas.com/docs/286317>) to place Enterprise Vault into a read-only and back into a read-write state, the batch file will not handle the integration needed for NetApp storage.

A modified version of this batch file has been placed in Appendix A which covers full integration with NetApp storage, along with NetApp SnapMirror and OSSV scheduling.

Note: The supplied batch file does require editing before it may be used. The batch file is divided into sections, with each section covering what modification is required based on your environment.

2.2.1 Network Time Protocol (NTP)

It is important that the NetApp storage system's time and time zone settings to the Enterprise Vault domain controller are synchronized.

If the time settings on the storage system and the domain controller are more than five minutes apart, the CIFS setup fails. The Kerberos protocol requires that the time settings on the storage system and domain controller be nearly the same. Once CIFS is running, if time drifts more than 15 minutes from the NetApp storage and the authenticating DC, Enterprise Vault services will not be able to connect to the CIFS share(s). To eliminate this problem, the best practice is to configure NetApp storage to use an NTP source.

To configure NetApp storage for NTP time synchronization, from a CLI session, issue the following commands:

```
NTAP> Options timed.enable on
NTAP> Options timed.log off
NTAP> Options timed.max_skew 5m
NTAP> Options timed.min_skew 0
NTAP> Options timed.protontp
NTAP> Options timed.sched hourly
NTAP> Options timed.servers <NTP source>
NTAP> Options timed.window 0s
```

Refer to the NetApp Data ONTAP system maintenance guide for detailed information on "timed" or from the NetApp console, type: "man na_rdate" for detailed instructions.

To temporarily adjust the date and time with an NTP server, type:

```
NTAP> rdate <NTP source>
```

2.2.2 Configuring SSH

On the NetApp storage, configure SecureAdmin™ to enable SSH2 to be able to send encrypted commands to NetApp storage.

Note: For customers who are using NetApp storage in a MultiStore configuration, NetApp fully supports SSH commands on vFiler0, but on all other virtual filers, the SSH command set is scaled down and does not support using the NetApp Snap command to vFilers beyond vFiler0.

SSH server needs a RSA host key and a DSA host key to support the SSH 2.0 protocol.

```
NTAP> secureadmin setup ssh
```

SSH Setup will now ask you for the sizes of the host and server keys.

For SSH 1.0 protocol, key sizes must be between 384 and 2048 bits.

For SSH 2.0 protocol, key sizes must be between 768 and 2048 bits.

The size of the host and server keys must differ by at least 128 bits.

Accept the defaults when it comes to selecting key size.

```
Please enter the size of host key for ssh1.x protocol [768] :
```

```
Please enter the size of server key for ssh1.x protocol [512] :
```

```
Please enter the size of host keys for ssh2.0 protocol [768] :
```

```
Is this correct? [yes]
```

Setup will now generate the host keys in the background. It will take a few minutes. A syslog message will be generated when Setup is complete.

```
NTAP> Wed Oct 25 05:59:56 GMT [rc:info]: SSH Setup: SSH Setup is done. Host keys are stored in /etc/sshd/ssh_host_key, /etc/sshd/ssh_host_rsa_key and /etc/sshd/ssh_host_dsa_key.
```

You can start SSH server with the command:

```
NTAP> secureadmin enable ssh2
```

Configure SSH

Search on the Internet with a search engine such as Google® and download both plink.exe and puttygen.exe. Create a directory called C:\SnapManager_Scripts. Place both programs in this folder. These two programs are required to setup the SSH keys. They are not required for the EVNTAP.BAT script.

```
SERVER> C:\SnapManager_Scripts> plink.exe <username>@<NetApp storage> ?
```

Here you can see the NetApp storage has accepted the connection and is now prompting for a password.

Next, generate keys for SSH with puttygen.exe. Be sure you DO NOT enter a passphrase when

generating the keys.

Begin by executing puttygen.exe:

- Select SSH-2 RSA radio button, located at the bottom middle of the GUI
- Accept the 1024 default for key size; the key size on the host does not have to match that of the NetApp storage, but it does have to be larger.
- Click Generate and you will be prompted to move your mouse in the key area.
- Once the Keys have been generated, select save public key and save private key. Save them to the directory c:\SnapManager_Scripts

Public key name = public_key

Private key name = private_key.ppk

Create an authorized_keys file. As a general rule the authorized_keys file is very sensitive; it does not want any line breaks. Do not edit this file with notepad; instead use wordpad or textpad.

When you open the public_key file generated by puttygen it will look like this.

```
---- BEGIN SSH2 PUBLIC KEY ----
Comment: "rsa-key-20070119"

AAAAB3NzaC1yc2EAAAABJQAAAIEAyQ8pESW3f2dRNNtnioOTPD0dyTVfW1TcIrFY
1aC/qMHH2AK9A5Kjd9dUBq7YudjakUiwZKvB7rucg7FaMbOZDqf/HvqdJf3Zem99
LaolDWBpGJRNqe8zmdWWnU/SXV9weWjsx6W+JeT9Urhfp/hbgidI8D6HxyJO/028
1Yro2XM=
---- END SSH2 PUBLIC KEY ----
```

You will need to strip all line breaks and extra text from this file to look like this. Note: After "ssh-rsa" there should be a space and then the key.

```
ssh-rsa
AAAAB3NzaC1yc2EAAAABJQAAAIEAyQ8pESW3f2dRNNtnioOTPD0dyTVfW1TcIrFY1aC/qMHH2AK9A5
Kjd9dUBq7YudjakUiwZKvB7rucg7FaMbOZDqf/HvqdJf3Zem99LaolDWBpGJRNqe8zmdWWnU/SXV9w
eWjsx6W+JeT9Urhfp/hbgidI8D6HxyJO/0281Yro2XM=
```

Notice that ssh-rsa is appended to the beginning of the file. Save this file as authorized_keys with no extension.

Create the following directory structure on the NetApp storage:

```
/etc/sshd/<username>/ .ssh
```


If you are doing this from a Windows host you will not be able to create the .ssh folder. Instead, create

```
/etc/sshd/<username>/ssh.
```

Copy the authorized_keys file into this directory; once copied use the mv command on the NetApp storage to rename the ssh directory:

```
NTAP> priv set advanced
NTAP*> mv /etc/sshd/<username>/ssh /etc/sshd/<username>/.ssh
NTAP*> priv set
```

Run plink.exe. on the Enterprise Vault server

```
SERVER> c:\SnapManager_Scripts> plink.exe -v -i
c:\SnapManager_Scripts\private_key.ppk <username>@<Storage Name> ?
```

The -v flag above is used to give a verbose output regarding the connection negotiation.

2.2.3 Preparing each Servers

The following software must be installed.

2.2.3.1 Enterprise Vault Server(s)

- Create a directory called C:\SnapManager_Scripts (this should have already been created in the SSH procedure)
- Create a sub folder called: C:\SnapManager_Scripts\LOG
- Install SSH - I use a freeware program called securessh.exe which can be found searching the Internet. You may prefer to use a supported program, such as Perl from ActiveState.
- Install NetApp Open Systems SnapVault® (OSSV) host agent to be able to Snapshot the registry and Enterprise Vault configuration files.

Place the following files in the C:\SnapMaanger_Scripts folder. These third-party tools are required to make the batch file script function. They are supported “as is” without any warranty or additional support from NetApp. Based on your application policy, you may wish to replace them with supported tools of similar functionality.

Tool	Download from	Version
PSEXEC.EXE	http://technet.microsoft.com	version 1.94, January 4, 2008
IgnoreArchiveBitTrigger.txt	http://seer.support.veritas.com/docs/273151.htm	
BLAT.EXE BLAT.DLL	http://www.blatt.net	version 2.6.2, February 25, 2007

BLAT.LIB		
EVNTAP.BAT	Located in Appendix A of this document	

2.2.3.2 Microsoft SQL Server(s)

- Follow the NetApp Deployment Guide for Symantec Enterprise Vault http://media.netapp.com/documents/tr_3635.pdf
- Ensure the correct Windows hot fixes, NetApp SnapDrive® and SnapManager for SQL® are installed following the best practice guideline located at <http://now.netapp.com>

2.2.4 Using EVNTAP.BAT for Cascading Snapshots

Use Windows Scheduled Tasks, and configure the EVNTAP.BAT file to run once per day. The default batch file will create a Snapshot each time it is run, retaining seven. If you desire a different retention, the :SNAPSHOT section needs to be changed in the batch file.

3 RESTORING EV TO A PREVIOUS SNAPSHOT

EVNTAP.BAT supports two switches, RESTORE and NORMAL.

To restore Enterprise Vault to a previously taken NetApp Snapshot, follow these steps:

1. Suspend the current Windows Scheduled Tasks of EVNTAP.BAT to ensure it does not attempt to run while a restore is in progress. (Properties of Scheduled Task, uncheck Enabled)
2. Manually run the EVNTAP.BAT file with the RESTORE switch. This places EV in a read-only mode
SERVER> EVNTAP.BAT RESTORE
3. From the Windows SQL server, launch SnapManager for SQL. Select Restore, followed by specifying the Snapshot you wish to restore. Choose Restore now.
4. From the NetApp storage GUI (http://<NetApp Storage>/na_admin) or the CLI, issue a SnapRestore of the EV Share volume. This is what is referenced in the script as %EVSHARE%. The date and time needs to match the restore used for the SnapManager for SQL. Note: As the Snapshot was created from a cascading Snapshot, meaning it occurred once the SnapManger for SQL had completed, the time may be different by a few minutes.
5. Once the restore has completed, run the batch file with the NORMAL switch, which indicates to EV to place EV back in read-write mode
SERVER> EVNTAP.BAT NORMAL
6. Release the Windows Task Scheduled of EVNTAP.BAT to enable the schedule

4 APPENDIX A – EVNTAP.BAT

```
@echo off
REM =====
REM EVNTAP.BAT
REM Framework was created by Symantec, December 13, 2006
REM Providing Read-Only / Read-Write support for Stand-alone, Microsoft Clustered and Veritas Clustered Servers
REM
REM Abhijit Lele, NetApp Professional Services, February 14, 2008
REM Added SQL Snapshot, SnapMirror, SnapVault, logging and email notification support
REM
REM Peter Henneberry, Global Systems Engineer, September 2008
REM Added Remote SQL, EV Rotating Snapshots, and Restore function support
REM =====

SET WDIR=C:\SnapManager_Scripts
SET PATH=%PATH%;C:\SnapManager_Scripts\;C:\Program Files\Enterprise Vault\;
SET WSSH=C:\Program Files\OpenSSH\bin\ssh.exe
SETLOCAL

REM =====
REM Set CLUSTER_TYPE to one of 'NONE', 'MSCS' or 'VCS'
REM =====
SET CLUSTER_TYPE=NONE

REM =====
REM Set the appropriate name for the Enterprise Vault SERVERNAME.
REM If in a cluster, set SERVERNAME to the appropriate virtual server name
REM =====
SET SERVERNAME=SERVERNAME

REM =====
REM The following setting is only needed for clustered Enterprise Vault installations.
REM Take extra care with VCS cluster names as they are case sensitive.
REM =====
SET CLUSTERNAME=YOURCLUSTERNAME

REM =====
REM The following setting is only needed for clustered Enterprise Vault installations.
REM For MSCS this should be the resource group for the virtual server.
```

REM For VCS this should be the service group for the virtual server. Take extra care as this is case sensitive.

REM =====

SET CLUSTERGROUP=YOURCLUSTERGROUP

REM =====

REM SQL server name, and SQL instance name, which is usually the Host name of the SQL server

REM SQLBKUP should point to the SQL server directory where NetApp SnapManager for SQL is installed

REM SQLADMIN & SQLPWD is the Domain name user and password to remotely access the computer via PSEXEC program

REM =====

SET SQLSRV=SQLSERVER

SET SQLINSTANCE=SQLINSTANCE

SET SQLBKUP=c:\program files\netapp\snapmanager for SQL server

SET SQLADMIN=administrator

SET SQLPWD=password

REM =====

REM SSH Private key for NetApp storage secure access

REM =====

SET SSHKEY=%WDIR%\private_key.ppk

REM =====

REM NetApp storage host name

REM Optional, NetApp storage host for SnapMirror

REM Optional, NetApp SnapMirror Volumes, from -> to

REM =====

SET NTAP=NETAPP1

SET NTAPMIR=NETAPP2

SET NTAPVOL1=EVINDEX

SET NTAPVOL2=EVINDEX2

SET NTAPVOL3=EVDATA

SET NTAPVOL4=EVDATA2

REM =====

REM NetApp storage root access for SSH

REM =====

SET NTAPUSER=root

REM =====

REM Enterprise Vault NetApp Share

REM =====

SET NTAPSHARE=EVSHARE

```

REM =====
REM OSSV volume on NetApp storage /vol/%OSSV%/EV_%SERVERNAME%
REM =====
SET OSSV=OSSVVOL

REM =====
REM BLAT email setup, modify the EMAILFROM, EMAILTO and EMAILSMTP as required
REM =====
SET LOGFILE=%WDIR%\LOG\backup.log
SET EMAILFROM=EV_%SERVERNAME%@EVTEST.NETAPP.COM
SET EMAILTO=SYSOP@EVTEST.NETAPP.COM
SET EMAILSMTP=EXCHANGE.EVTEST.NETAPP.COM
DEL %LOGFILE%

REM =====
REM List all services here that are to be restarted when setting Enterprise Vault into read-only or read-write
REM mode. Take extra care with VCS because the generic service resource names are case sensitive.
REM =====
SET SERVICES_LIST=
SET SERVICES_LIST=%SERVICES_LIST% "Enterprise Vault Task Controller Service"
SET SERVICES_LIST=%SERVICES_LIST% "Enterprise Vault Storage Service"
SET SERVICES_LIST=%SERVICES_LIST% "Enterprise Vault Indexing Service"
SET SERVICES_LIST=%SERVICES_LIST% "Enterprise Vault Shopping Service"

GOTO :MAIN

:SETEVREG
REM =====
CALL :DOCMD_NUL reg add "\\%1\HKLM\Software\KVS\Enterprise Vault\Storage" /f /v EnableArchive /t REG_DWORD /d %2
CALL :DOCMD_NUL reg add "\\%1\HKLM\Software\KVS\Enterprise Vault\Storage" /f /v EnableCrawler /t REG_DWORD /d %2
CALL :DOCMD_NUL reg add "\\%1\HKLM\Software\KVS\Enterprise Vault\Storage" /f /v EnableExpiry /t REG_DWORD /d %2
CALL :DOCMD_NUL reg add "\\%1\HKLM\Software\KVS\Enterprise Vault\Storage" /f /v EnableFileWatch /t REG_DWORD /d %2
CALL :DOCMD_NUL reg add "\\%1\HKLM\Software\KVS\Enterprise Vault\Storage" /f /v EnablePSTMigrations /t REG_DWORD /d %2
CALL :DOCMD_NUL reg add "\\%1\HKLM\Software\KVS\Enterprise Vault\Storage" /f /v EnableReplayIndex /t REG_DWORD /d %2
CALL :DOCMD_NUL reg add "\\%1\HKLM\Software\KVS\Enterprise Vault\Storage" /f /v EnableRestore /t REG_DWORD /d %2
CALL :DOCMD_NUL reg add "\\%1\HKLM\Software\KVS\Enterprise Vault\Storage" /f /v EnablePSTMigrations /t REG_DWORD /d %2
GOTO :END

:EVSERVICE_START
REM =====

```

```

IF x%1 EQU x GOTO :END
CALL :DOCMD EVService start %SERVERNAME% %1
SHIFT
GOTO :EVSERVICE_START

:EVSERVICE_STOP
REM =====
IF x%1 EQU x GOTO :END
CALL :DOCMD EVService stop %SERVERNAME% %1
SHIFT
GOTO :EVSERVICE_STOP

:MSCS_ONLINE
REM =====
IF x%1 EQU x GOTO :END
REM Remove spaces and quotes from the service name and from the resource name
SET RESNAME=%1
SET RESNAME=%RESNAME:=%
SET RESNAME=%CLUSTERGROUP%-%RESNAME:~1,-1%
CALL :DOCMD cluster /cluster:%CLUSTERNAME% resource "%RESNAME%" /online
SHIFT
GOTO :MSCS_ONLINE

:MSCS_OFFLINE
REM =====
IF x%1 EQU x GOTO :END
REM Remove spaces and quotes from the service name and from the resource name
SET RESNAME=%1
SET RESNAME=%RESNAME:=%
SET RESNAME=%CLUSTERGROUP%-%RESNAME:~1,-1%
CALL :DOCMD cluster /cluster:%CLUSTERNAME% resource "%RESNAME%" /offline
SHIFT
GOTO :MSCS_OFFLINE

:VCS_ONLINE
REM =====
IF x%1 EQU x GOTO :END
REM Remove spaces and quotes from the service name and from the resource name
SET RESNAME=%1
SET RESNAME=%RESNAME:=%
SET RESNAME=%CLUSTERGROUP%-%RESNAME:~1,-1%

```

```

CALL :GET_VCS_LASTONLINE %CLUSTERNAME% %RESNAME%
IF x%VCS_LASTONLINE% EQU x ECHO %RESNAME% not onlined - cannot get VCS LastOnline attribute && SHIFT && GOTO :VCS_ONLINE
CALL :DOCMD hares -online "%RESNAME%" -sys %VCS_LASTONLINE%
CALL :DOCMD hares -wait "%RESNAME%" State ONLINE -sys %VCS_LASTONLINE%
SHIFT
GOTO :VCS_ONLINE

:VCS_OFFLINE
REM =====
IF x%1 EQU x GOTO :END
REM Remove spaces and quotes from the service name and from the resource name
SET RESNAME=%1
SET RESNAME=%RESNAME:=%
SET RESNAME=%CLUSTERGROUP%-%RESNAME:~1,-1%
CALL :GET_VCS_LASTONLINE %CLUSTERNAME% %RESNAME%
IF x%VCS_LASTONLINE% EQU x ECHO %RESNAME% not offlined - cannot get VCS LastOnline attribute && SHIFT && GOTO :VCS_OFFLINE
CALL :DOCMD hares -offline "%RESNAME%" -sys %VCS_LASTONLINE%
CALL :DOCMD hares -wait "%RESNAME%" State OFFLINE -sys %VCS_LASTONLINE%
SHIFT
GOTO :VCS_OFFLINE

:GET_VCS_LASTONLINE
REM =====
SET VCS_CLUSTER=%1
SET VCS_RES=%2
SET VCS_LASTONLINE=
for /F "tokens=1,2,3* delims=" %%A IN ('hares -display "%VCS_RES%" -clus "%VCS_CLUSTER%") DO CALL :FIND_VCS_LASTONLINE_ATTRIBUTE
%%VCS_RES% %%VCS_CLUSTER% %%A %%B %%C %%D
GOTO :END

:FIND_VCS_LASTONLINE_ATTRIBUTE
REM =====
IF "%1" NEQ "%3" GOTO :END
IF "%2" NEQ "%5" GOTO :END
IF "%4" NEQ "LastOnline" GOTO :END
SET VCS_LASTONLINE=%6
GOTO :END

:DOCMD
REM =====
set EXCMD=%*

```

```
IF "%NOEXECUTE%" EQU "TRUE" echo %EXCMD%
IF "%NOEXECUTE%" NEQ "TRUE" %EXCMD%
GOTO :END
```

```
:DOCMD_NUL
REM =====
set EXCMD=%*
IF "%NOEXECUTE%" EQU "TRUE" echo %EXCMD%
IF "%NOEXECUTE%" NEQ "TRUE" %EXCMD% > nul
GOTO :END
```

```
:MAIN
REM =====
SET RESTORE=FALSE
IF /I "%1" EQU "restore" SET RESTORE=TRUE
SET NORMAL=FALSE
IF /I "%1" EQU "normal" SET NORMAL=TRUE
IF EXIST %wDIR%\EV.LOCK (
EVENTCREATE /T ERROR /ID 1 /D "EV Backup failed due to another instance of EV Backup already running on %SERVERNAME%" /SO NetAppSMEV /L
Application
ECHO EV Backup failed due to another instance of EV Backup already running on %SERVERNAME% >> %LOGFILE%
ECHO TIMESTAMP: %DATE% , %TIME% >> %LOGFILE%
GOTO :SEND_EMAIL
EXIT 1
)
IF "%CLUSTER_TYPE%" EQU "MSCS" GOTO :MSCS
IF "%CLUSTER_TYPE%" EQU "VCS" GOTO :VCS
```

```
REM Non cluster
SET SERVICE_START_FN=EVSERVICE_START
SET SERVICE_STOP_FN=EVSERVICE_STOP
GOTO :MAIN_CONTINUE
```

```
:MSCS
REM =====
SET SERVICE_START_FN=MSCS_ONLINE
SET SERVICE_STOP_FN=MSCS_OFFLINE
GOTO :MAIN_CONTINUE
```

```
:VCS
REM =====
```



```

SET SERVICE_START_FN=VCS_ONLINE
SET SERVICE_STOP_FN=VCS_OFFLINE
GOTO :MAIN_CONTINUE

:MAIN_CONTINUE
REM =====
ECHO "XXXXXXXX" > %wDIR%\EV.LOCK
IF "%NORMAL%" == "TRUE" (
ECHO Placing EV in Read-Write mode following a Snapshot restore >> %LOGFILE%
ECHO TIMESTAMP: %DATE%, %TIME% >> %LOGFILE%
GOTO :READWRITE_MODE
)

:READONLY_MODE
REM =====
CALL :%SERVICE_STOP_FN% %SERVICES_LIST%
CALL :SETEVREG %SERVERNAME% 0
CALL :%SERVICE_START_FN% %SERVICES_LIST%
IF "%RESTORE%" == "TRUE" (
ECHO Updated registry for Enterprise Vault read-only mode >> %LOGFILE%
ECHO TIMESTAMP: %DATE%, %TIME% >> %LOGFILE%
GOTO :END
)

:SQLBKUP
REM =====
ECHO Backing up SQL Server Including mirror FOR %SQLBKUP% >> %LOGFILE%
ECHO TIMESTAMP: %DATE%, %TIME% >> %LOGFILE%
%WDIR%\psexec \\%SQLSRV% -u %SQLADMIN% -p %SQLPWD% -w "%SQLBKUP%" "%SQLBKUP%\smsqlbi.exe" -H "%SQLINSTANCE%" -D 8 -R -G -S
"%SQLINSTANCE%" -C 0 -Q "%SQLINSTANCE%"
IF %ERRORLEVEL% NEQ 0 (
ECHO FAILED to Backup SQL Server %SQLBKUP% >> %LOGFILE%
ECHO TIMESTAMP: %DATE%, %TIME% >> %LOGFILE%
EVENTCREATE /T ERROR /ID 1 /D "Failed to Backup on SQL server %SQLSRV%" /SO NetAppSMEV /L Enterprise Vault
) ELSE (
ECHO Successfully completed SQL server backup %SERVERNAME% >> %LOGFILE%
ECHO TIMESTAMP: %DATE%, %TIME% >> %LOGFILE%
)

:SNAPSHOT
ECHO Taking a NetApp Snapshot of Enterprise Vault share (maintain 7) >> %LOGFILE%

```

```

%WSSH% -i %SSHKEY% %NTAPUSER%@%NTAP% snap delete -V %NTAPSHARE% %NTAPSHARE%.daily.7
%WSSH% -i %SSHKEY% %NTAPUSER%@%NTAP% snap rename -V %NTAPSHARE% %NTAPSHARE%.daily.6 %NTAPSHARE%.daily.7
%WSSH% -i %SSHKEY% %NTAPUSER%@%NTAP% snap rename -V %NTAPSHARE% %NTAPSHARE%.daily.5 %NTAPSHARE%.daily.6
%WSSH% -i %SSHKEY% %NTAPUSER%@%NTAP% snap rename -V %NTAPSHARE% %NTAPSHARE%.daily.4 %NTAPSHARE%.daily.5
%WSSH% -i %SSHKEY% %NTAPUSER%@%NTAP% snap rename -V %NTAPSHARE% %NTAPSHARE%.daily.3 %NTAPSHARE%.daily.4
%WSSH% -i %SSHKEY% %NTAPUSER%@%NTAP% snap rename -V %NTAPSHARE% %NTAPSHARE%.daily.2 %NTAPSHARE%.daily.3
%WSSH% -i %SSHKEY% %NTAPUSER%@%NTAP% snap rename -V %NTAPSHARE% %NTAPSHARE%.daily.1 %NTAPSHARE%.daily.2
%WSSH% -i %SSHKEY% %NTAPUSER%@%NTAP% snap create -V %NTAPSHARE% %NTAPSHARE%.daily.1
ECHO Successfully completed taking a NetApp Snapshot of Enterprise Vault share >> %LOGFILE%
ECHO TIMESTAMP: %DATE% , %TIME% >> %LOGFILE%

```

:SNAPMIRROR

```

REM If you are using NetApp SnapMirror, make the appropriate changes
REM $WSSH% -i %SSHKEY% %NTAPUSER%@%NTAP% snapmirror update -S %NTAPMIR%:%NTAPVOL1% %NTAPVOL2%
REM $WSSH% -i %SSHKEY% %NTAPUSER%@%NTAP% snapmirror update -S %NTAPMIR%:%NTAPVOL3% %NTAPVOL4%
REM ECHO Successfully initiated a SnapMirror update on %NTAPMIR% >> %LOGFILE%
REM ECHO TIMESTAMP: %DATE% , %TIME% >> %LOGFILE%

```

:READWRITE_MODE

```

CALL :%SERVICE_STOP_FN% %SERVICES_LIST%
CALL :SETEVREG %SERVERNAME% 1
CALL :%SERVICE_START_FN% %SERVICES_LIST%

```

:SAFETYCOPY

```

REM =====
REM If EV is configured with Safety Copy, to remove the safety copy following a successful backup
REM copy %WDIR%\IgnoreArchiveBitTrigger.txt \\%NTAP%\%NTAPSHARE%\<Folder Name>\IgnoreArchiveBitTrigger.txt
REM NET STOP %SERVERNAME% "Enterprise Vault Storage Service" /y
REM NET START %SERVERNAME% "Enterprise Vault Storage Service"
REM ECHO Restoring the Safety Copy file >> %LOGFILE%
REM ECHO TIMESTAMP: %DATE% , %TIME% >> %LOGFILE%

```

:OSSV

```

REM =====
REM NetApp's OSSV should be launched at this point to perform a backup
REM on each EV server's to capture the registry and configuration files.
REM %WSSH% -i %SSHKEY% %NTAPUSER%@%NTAP% snapvault update %NTAP%:/vol/%OSSV%/EV_%SERVERNAME%
REM ECHO Successfully initiated an OSSV update on %NTAP% >> %LOGFILE%
REM ECHO TIMESTAMP: %DATE% , %TIME% >> %LOGFILE%

```

:SEND_EMAIL

```
%WDIR%\Blat.exe %LOGFILE% -to %EMAILTO% -f %EMAILFROM% -server %EMAILSMTP% -attach "C:\Windows\System32\config\EVEventLog.evt"  
DEL %wDIR%\EV.LOCK
```

```
:END
```

5 BEST PRACTICES

1. In general, with Volume SnapMirror (VSM) the best practice is to schedule a close of the Symantec vault partition quarterly, and open a new one on a new volume; and then start SnapMirror or SnapVault replication for disaster recovery. Based on the amount of data, this schedule may need to be adjusted to allow for easier management of EV's partition growth, i.e. monthly.
2. Always place data within a NetApp Qtree, which resides within a volume. This provides for better management of SnapMirror restores. Unlike traditional SnapMirror, the Qtree SnapMirror (QSM) will not delete any Snapshots on the intended destination. For further information on QSM, refer to NetApp's Data ONTAP 7.3 Administration guide.

6 CONCLUSION

The combination of Symantec Enterprise Vault with NetApp storage systems provides customers the capability to archive and protect business-critical e-mail in a simple-to-manage unified storage environment. The procedures made in this paper are intended to be an overview of the Enterprise Vault backup and recovery architecture. Although the EVNTAP.BAT script will work for most environments, to ensure all components have been properly configured and the best practices have been followed for Enterprise Vault, Microsoft SQL, and NetApp, a customer should consider involving NetApp Professional Services or a NetApp certified partner to perform a site evaluation and health check of each application prior to the deployment of the EVNTAP.BAT script



© 2008 NetApp. All rights reserved. Specifications are subject to change without notice. NetApp, the NetApp logo, Go further, faster, DataFabric, Data ONTAP, FilerView, FlexVol, MultiStore, NOW, ONTAPI, RAID-DP, SecureAdmin, SecureShare, SnapDrive, SnapManager, SnapMirror, SnapRestore, Snapshot, vFiler, VFM, Virtual File Manager, and WAFL are trademarks or registered trademarks of NetApp, Inc. in the United States and/or other countries. Microsoft, Windows, Windows NT, and SharePoint are registered trademarks and SQL Server is a trademark of Microsoft Corporation. Linux is a registered trademark of Linus Torvalds. Java and Sun are trademarks of Sun Microsystems, Inc. Oracle is a registered trademark of Oracle Corporation. Symantec is a trademark of Symantec Corporation or its affiliates in the U.S. and other countries. SAP is a registered trademark of SAP AG. UNIX is a registered trademark of The Open Group. All other brands or products are trademarks or registered trademarks of their respective holders and should be treated as such.